Explore 3D Figures

Dr. Jing Wang

(517)2675965, wangj@lcc.edu

Lansing Community College, Michigan, USA

Part I. 3D Modeling

In this part, we create 3D models using *Mathematica* for various solids in 3D space, and then we print out those solids using a 3D printer.

Background: Three dimensional coordinate systems: Cartesian (rectangular), cylindrical (polar), and spherical coordinates. Parametric surfaces.

1.1. Examples

This section provides you with ten examples of modeling 3D solids using *Mathematica.* You will see that for some regions, they can be modeled using different methods, but for some other regions, they can only be modeled using one method or using parametric equations.

General Regions

Example 1.1. Create a 3D model for the solid that lies under the surface $z = x y^2$ and over the triangular region described by $0 \leq x \leq 1$ and $x \leq y \leq 3x$.

This solid can be modeled using two methods.

Method 1: Using RegionPlot3D to graph the solid.

```
exampleTria = RegionPlot3D[
  z < x * y^2 && 0 < x < 1 && x < y < 3 x,
  {x, 0, 1}, {y, 0, 3}, {z, 0, 9},
  Mesh → None,
  PlotPoints → 100,
  {Axes → False, Boxed → False},
  ImageSize → 200]
```



```
Export["C:\\Users\\wangj\\Desktop\\example1.1a.stl", exampleTria]
```

```
C:\Users\wangj\Desktop\example1.1a.stl
```

Method 2: Using Plot3D to graph the solid.

```
exampleTrib = Plot3D[x * y^2, {x, 0, 1}, {y, 0, 3}, BoxRatios → {1, 1, 1},
  RegionFunction → Function[{x, y}, x ≤ y ≤ 3 x && x ≤ 1],
  Filling → Bottom,
   ViewPoint → {4, -3, 2},
  Mesh → None,
  {Axes → False, Boxed → False}, ImageSize → 200]
```



```
Export["C:\\Users\\wangj\\Desktop\\example1.1b.stl", exampleTrib]
```

```
C:\Users\wangj\Desktop\example1.1b.stl
```

Example 1.2. Create a 3D model for the intersection between two cylinders: $y^2 + z^2 = 1$ and $x^2 + z^2 = 1$.

This solid can be better modeled using RegionPlot3D.

```
twoCylinders = RegionPlot3D[ y^2 + z^2 < 1 && x^2 + z^2 < 1 ,
  {x, -1.5, 1.5}, {y, -1.5, 1.5}, {z, -1.5, 1.5}, PlotPoints → 100,
  Mesh → False, Axes → False, Boxed → False, ImageSize → 200]
```



```
Export["C:\\Users\\wangj\\Desktop\\twoCylinders.stl", twoCylinders]
```

```
C:\Users\wangj\Desktop\twoCylinders.stl
```

Example 1.3. Create a 3D model for the the solid that lies under the surface $z = (x - y)^2 / 2$ and over the sector of the unit disk in which $-\pi/4 \le \theta \le \pi/4$.

Here we create this model using Plot3D.

```
exampleSect = Plot3D[(x - y) ^2 / 2 , {x, 0, 1}, {y, -0.75, 0.75},
  RegionFunction → Function[{x, y}, -π / 4 ≤ ArcTan[y / x] ≤ π / 4 && x^2 + y^2 ≤ 1],
  Filling → Bottom,
  ViewPoint → {3, 4, 3},
  Mesh → None,
  {Axes → False, Boxed → False}, ImageSize → 200]
```



```
Export["C:\\Users\\wangj\\Desktop\\example1.3.stl", exampleSect]
```

```
C:\Users\wangj\Desktop\example1.3.stl
```

Example 1.4. Create a 3D model for the solid that lies under the surface $z = (x^2 + 1)\, y$ and over the disk bounded by the circle $x^2 + (y - 1)^2 = 1$.

Here we create this model using Plot3D.

```
exampleDisk = Plot3D[(x^2 + 1) * y , {x, -1.1, 1.1}, {y, -0.1, 2.1},
  RegionFunction → Function[{x, y}, x^2 + (y - 1) ^2 ≤ 1],
  Filling → Bottom,
  Mesh → None,
  {Axes → False, Boxed → False},
  ImageSize → 200, ViewPoint → {-3, -3, 2}]
```



```
Export["C:\\Users\\wangj\\Desktop\\example1.4.stl", exampleDisk]
```

```
C:\Users\wangj\Desktop\example1.4.stl
```

Example 1.5. Create a 3D model for the solid that occupies the region bounded by the paraboloid $z = 1 + x^2 + y^2$, the plane $x + y + z = 0$, and the cylinder $x^2 + y^2 = 1$.

For this solid, it’s easier to use RegionPlot3D.

```
intersection = RegionPlot3D[
  -x - y < z < 1 + x^2 + y^2 && -1 < x < 1 && -Sqrt[1 - x^2] < y < Sqrt[1 - x^2],
  {x, -1, 1}, {y, -1, 1}, {z, -2, 2},
  Mesh → None,
  PlotPoints → 100,
  {Axes → False, Boxed → False},
  ImageSize → 200]
```



```
Export["C:\\Users\\wangj\\Desktop\\example1.5.stl", intersection]
```

```
C:\Users\wangj\Desktop\example1.5.stl
```

Example 1.6. Create a 3D model for the the solid that lies Inside the upper hemisphere $x^2 + y^2 + z^2 = 1$ and outside the cone $z = \sqrt{x^2 + y^2}$.

```
partOfhemisphere = RegionPlot3D[x^2 + y^2 + z^2 < 1
   && 0 < z^2 < x^2 + y^2
   && z > 0,
  {x, -1, 1},
  {y, -1, 1},
  {z, -1, 1},
  PlotPoints → 100,
  Mesh → None,
  {Axes → False, Boxed → False},
  ImageSize → 200]
```



```
Export["C:\\Users\\wangj\\Desktop\\partOfhemisphere.stl", partOfhemisphere]
```

```
C:\Users\wangj\Desktop\partOfhemisphere.stl
```

Example 1.7. A Star Fruit. Create a 3D model for the the region inside the spherical surface $\rho = 1 + 0.2\,\mathrm{Sin}[5\,\theta]$, $0 \le \phi \le \pi$, $0 \le \theta \le 2\,\pi$.

The surface of the solid is described in spherical coordinates. It’s easier to model using SphericalPlot3D.

```
starFruit = SphericalPlot3D[1 + 0.2 Sin[5 θ], {ϕ, 0, π}, {θ, 0, 2 π},
          PlotStyle → Directive[Orange, Opacity[0.7], Specularity[White, 10]],
          Mesh → None, PlotPoints → 50,
          {Axes → False, Boxed → False}, ImageSize → 200]
```



```
Export["C:\\Users\\wangj\\Desktop\\starFruit.stl", starFruit]
```

```
C:\Users\wangj\Desktop\starFruit.stl
```

Regions Described by Parametric Equations

Example 1.8. A Torus (Donut). Create a 3D model for the region inside the surface given by the parametric equations: $x = (2 + \cos\alpha)\cos\theta$, $y=(2 + \cos\alpha)\sin\theta$, $z=\sin\alpha$, $0 \le \theta \le 2\pi$, $0 \le \theta \le 2\pi$.

The surface of the solid is described by parametric equations. It’s modeled using ParametricPlot3D.

```
torus = ParametricPlot3D[
   {(2 + Cos[α]) Cos[θ], (2 + Cos[α]) Sin[θ], Sin[α]}, {α, 0, 2 π}, {θ, 0, 2 π},
       Mesh → None, PlotPoints → 50,
       {Axes → False, Boxed → False}, ImageSize → 200]
```



```
Export["C:\\Users\\wangj\\Desktop\\torus.stl", torus]
```

```
C:\Users\wangj\Desktop\torus.stl
```

Example 1.9. Make a 3D solid that remains when a hole of radius 2 is drilled through the center of a sphere of radius 4.

Note: Here we use ParametricPlot3D for two surfaces: the sphere and the hole. They both are described using parametric surfaces.

```
aHole = ParametricPlot3D[{{√(16 - z^2) * Cos[θ], √(16 - z^2) * Sin[θ], z},
    {2 * Cos[θ], 2 * Sin[θ], z}}, {θ, 0, 2 π}, {z, -√12, √12},
  Mesh → None,
  {Axes → False, Boxed → False},
  ImageSize → 200]
```



```
Export["C:\\Users\\wangj\\Desktop\\aHole.stl", aHole]
```

```
C:\Users\wangj\Desktop\aHole.stl
```

Example 1.10. An Ice Cream Cone: Make a 3D model for the region that lies below the sphere $x^2 + y^2 + z^2 = z$ and above the cone $z = \sqrt{x^2 + y^2}$.

This model is created using ParametricPlot3D for two surfaces.

```
iceCreamCone = ParametricPlot3D[
  {{Cos[ϕ] * Sin[ϕ] * Cos[θ], Cos[ϕ] * Sin[ϕ] * Sin[θ], Cos[ϕ] * Cos[ϕ]},
   { Cos[ϕ] * Sin[ϕ] * Cos[θ], Cos[ϕ] * Sin[ϕ] * Sin[θ], Cos[ϕ] * Sin[ϕ] }} , {ϕ,
   0, π/4}, {θ, 0, 2 π},
  Mesh → None,
  PlotPoints → 100,
  {Axes → False, Boxed → False},
  ImageSize → 200]
```



```
Export["C:\\Users\\wangj\\Desktop\\iceCreamCone.stl", iceCreamCone]
```

```
C:\Users\wangj\Desktop\iceCreamCone.stl
```

1.2. Exercises

This section provides you with ten exercise problems. For the first nine problems, you are asked to create 3D models for the given solids in space. For the last problem, you are asked to create your own 3D model(s).

Problem 1.1. Make a 3D model for the solid that lies under the surface $z = x^2 y^2$ and over the region described by 0≤x≤Sin[y] and $0 \le y \le \pi$.

Problem 1.2. Make a 3D model for the solid occupies the region that lies under the plane $z = x - y/4$ and above the planar region between the graphs $y = \pm$Sin[πx] for $0 \le x \le 1$.

Problem 1.3. Make a 3D model for the solid that lies under the surface $z = 1 - 4x^2 y^2$ and over the unit disk, i.e. the region inside the unit circle.

Problem 1.4. Make a solid that lies under the surface $z = 5y$ and above the sector of the unit disk in which $0 \le \theta \le \pi$.

Problem 1.5. A Bumpy Sphere: Make a 3D model for the region inside the spherical surface $\rho = 1 + 0.2\,\text{Sin}[5\,\phi] * \text{Sin}[3\,\theta]$, $0 \le \phi \le \pi$, $0 \le \theta \le 2\pi$.

Problem 1.6. Make a 3D model for the intersection region between two spheres given by the equations: $x^2 + y^2 + z^2 = 1$ and $x^2 + y^2 + (z-1)^2 = 1$.

Problem 1.7. Make a 3D model for the solid that occupies the region bounded by the unit sphere $x^2 + y^2 + z^2 = 1$ and the cone $z = \sqrt{x^2 + y^2}$.

Problem 1.8. Make a 3D model for the solid that occupies the region bounded by the surface given by the equation $\rho = \phi\,\text{Sin}[\theta/2]$ in spherical coordinates.

Problem 1.9. Make a 3D model for the solid that occupies the region bounded by the surface given by the equation $\rho = 1 + \text{Sin}[\phi]$ in spherical coordinates.

Problem 1.10. Create your own 3D model(s) and give it (them) a name(s).

Part II. Volumes and Surface Areas

In this second part, we compute volumes and surface areas for some of the solids discussed in part I.
Background: Multiple integrals and solids of revolution.

2.1. Examples

This section provides you with various examples of computing volumes or surface areas for the solids discussed in Part I. Note: For some regions, there are more than one way to compute volumes or surface areas.

Integrals in Cartesian Coordinates

Example 2.1. Find the exact volume of the solid that lies under the surface $z = x\,y^2$ and over the triangular region described by $0 \le x \le 1$and $x \le y \le 3\,x$.

We first visualize the solid.

```
RegionPlot3D[
 z < x * y^2 && 0 < x < 1 && x < y < 3 x,
 {x, 0, 1}, {y, 0, 3}, {z, 0, 9},
 PlotPoints → 100,
 Mesh → None,
 {Axes → True, Boxed → True},
 ImageSize → 200]
```



Method 1. Here we compute the volume of the solid by evaluating the double integral for the function $f(x, y) = xy^2$ over the domain region $D = \{(x, y),\ \ 0 \le x \le 1,\ \ x \le y \le 3\,x\}$.

```
volumeTri = ∫_0^1 ∫_x^(3 x) x y^2 ⅆy ⅆx
```

$\frac{26}{15}$

```
N[volumeTri]
```

1.73333

Method 2. We can also compute the volume of the solid by evaluating the triple integral as follows.

$$\int_0^1 \int_x^{3x} \int_0^{x y^2} 1 \, dz \, dy \, dx$$

$\frac{26}{15}$

Example 2.2. Find the area of the part of the surface $z = x y^2$ that lies above the triangular region described by $0 \le x \le 1$and $x \le y \le 3x$. (This is the top of the solid in example 2.1.)

Here is a look of the top of the solid.

```
Plot3D[x * y^2, {x, 0, 1}, {y, 0, 3}, BoxRatios → {1, 1, 1},
 RegionFunction → Function[{x, y}, x ≤ y ≤ 3 x && x ≤ 1],
 Filling → Bottom,
  ViewPoint → {4, -3, 2},
 {Axes → True, Boxed → True}, ImageSize → 200]
```



Since the top surface is the graph of the function: $f(x, y) = x y^2$, to compute the area, we use the formula $\iint_D \sqrt{f_x^2 + f_y^2 + 1} \, dA$.

```
Clear[f];
f[x_, y_] := x y^2
{fx[x_, y_] = ∂x f[x, y], fy[x_, y_] = ∂y f[x, y]}
```

$\{y^2, 2\,x\,y\}$

Here is the integrand.

```
integrand = Sqrt[fx[x, y]^2 + fy[x, y]^2 + 1]
```

$\sqrt{1 + 4\,x^2\,y^2 + y^4}$

Mathematica will not be successful in finding the exact value of the integral; so we will just ask for a numerical approximation with NIntegrate.

```
areaTri = NIntegrate[integrand, {x, 0, 1}, {y, x, 3 x}]
```

3.24923

Example 2.3. Find the volume of the solid that occupies the region bounded by the intersection of two cylinders: $y^2 + z^2 = 1$ and $x^2 + z^2 = 1$.

Here is a picture of the solid.

```
twoCylinders =
 RegionPlot3D[ y^2 + z^2 < 1 && x^2 + z^2 < 1 , {x, -1.5, 1.5}, {y, -1.5, 1.5},
  {z, -1.5, 1.5}, PlotPoints → 100, Mesh -> None, Axes → True, ImageSize → 200]
```



Note the symmetry of the solid! The volume of the solid is 16 times the volume of the part in the first octant given by the double integral of $f(x, y) = \sqrt{1 - x^2}$ over the domain region D={(x,y), 0≤x≤1, 0≤y≤x}.

```
volumeCylinders = 16 ∫_0^1 ∫_0^x Sqrt[1 - x^2] dy dx
```

$\frac{16}{3}$

```
N[volumeCylinders]
```

```
5.33333
```

Integrals in Cylindrical (Polar) Coordinates

Example 2.4. Find the volume of the solid that lies under the surface $z = (x - y)^2 / 2$ and over the sector of the unit disk in which $-\pi/4 \leq \theta \leq \pi/4$.

Here is a picture of the solid.

```
Plot3D[(x - y)^2/2, {x, 0, 1}, {y, -0.75, 0.75},
 RegionFunction → Function[{x, y}, -π/4 ≤ ArcTan[y/x] ≤ π/4 && x^2 + y^2 ≤ 1],
 Filling → Bottom,
 Mesh → None,
 {Axes → True, Boxed → True},
 ImageSize → 200, ViewPoint → {3, 4, 3}]
```



To evaluate the volume, we use a double integral in polar coordinates. We rewrite the function in polar coordinates. Note the “extra” factor r in the integrand.

```
volumeWedge = ∫_{-π/4}^{π/4} ∫_0^1 (r * Cos[θ] - r * Sin[θ])^2 / 2 * r ⅆr ⅆθ
```

$$\frac{\pi}{16}$$

```
N[volumeWedge]
```

```
0.19635
```

Example 2.5. Find the area of the part of the surface $z = (x - y)^2 / 2$ that lies above the sector of the unit disk in which $-\pi/4 \leq \theta \leq \pi/4$. (This is the top surface of the solid in example 2.4.)

Here is a picture of the top of the solid.

```
Plot3D[(x - y)^2/2, {x, 0, 1}, {y, -0.75, 0.75},
 RegionFunction → Function[{x, y}, -π/4 ≤ ArcTan[y/x] ≤ π/4 && x^2 + y^2 ≤ 1],
 Filling → Bottom,
 {Axes → True, Boxed → True},
 ImageSize → 200, ViewPoint → {3, 4, 3}]
```



Since the top surface is the graph of the function: $f(x, y) = (x - y)^2 / 2$, to compute the surface area, we can use the formula $\iint_D \sqrt{f_x^2 + f_y^2 + 1}\ dA$.

```
Clear[f]
```

```
f[x_, y_] := (x - y)^2/2;
{fx[x_, y_] = ∂x f[x, y], fy[x_, y_] = ∂y f[x, y]}
```

```
{x - y, -x + y}
```

It will be convenient to use polar coordinates. Let's have a look at the function we want to integrate (including the extra r).

```
integrand = r √(fx[r Cos[θ], r Sin[θ]]^2 + fy[r Cos[θ], r Sin[θ]]^2 + 1) // Simplify
```

$$r \sqrt{1 + 2 r^2 - 2 r^2 \mathrm{Sin}[2 \theta]}$$

Mathematica will not be successful in finding the exact value of the integral; so we'll just ask for a numerical approximation with: NIntegrate.

```
NIntegrate[integrand, {r, 0, 1}, {θ, -π/4, π/4}]
```

```
1.0794
```

Example 2.6. Find the volume of the solid that lies under the surface $z = (x^2 + 1)\, y$ and over the disk bounded by the circle $x^2 + (y - 1)^2 = 1$.

Here is a picture of the solid.

```
Plot3D[(x^2 + 1) * y , {x, -1.1, 1.1}, {y, -0.1, 2.1},
 RegionFunction → Function[{x, y}, x^2 + (y - 1)^2 ≤ 1],
 Filling → Bottom,
 Mesh → None,
 {Axes → True, Boxed → True},
 ImageSize → 200, ViewPoint → {-3, -3, 2}]
```



To evaluate the volume, we use a double integral in polar coordinates.
First we convert the equation of the circle $x^2 + (y-1)^2 = 1$ in polar coordinate.

```
eqn := x² + (y - 1)² == 1
```

notice in polar coordinates this becomes

```
eqn /. {x → r * Cos[θ], y → r * Sin[θ]} // Simplify
```

```
r² == 2 r Sin[θ]
```

Dividing each side by r reveals that $r = 2\,\text{Sin}[\theta]$ is the polar equation of the circle. Hence the domain region is described by $0 \leq \theta \leq \pi$ and $0 \leq r \leq 2\,\text{Sin}[\theta]$.
So the integral that gives the volume is computed as follows.

```
f[x_, y_] = (x² + 1) y;
```

$$\text{volume6} = \int_0^{\pi}\int_0^{2\,\text{Sin}[\theta]} f[r\,\text{Cos}[\theta],\, r\,\text{Sin}[\theta]]\; r\, dr\, d\theta$$

$$\frac{5\pi}{4}$$

```
N[volume6]
```

```
3.92699
```

Integrals in Spherical Coordinates

Example 2.7. Find the volume of the solid that lies inside the upper hemisphere $x^2 + y^2 + z^2 = 1$ and outside the cone $z = \sqrt{x^2 + y^2}$.

Here a picture of the solid.

```
RegionPlot3D[x^2 + y^2 + z^2 < 1
  && 0 < z^2 < x^2 + y^2
  && z > 0,
 {x, -1, 1},
 {y, -1, 1},
 {z, -1, 1},
 PlotPoints → 70,
 Mesh → None,
 AxesLabel → Automatic,
 ImageSize → 200]
```



The solid is easier to describe in spherical coordinates: $0 \le \rho \le 1$, $\pi/4 \le \phi \le \pi/2$, $0 \le \theta \le 2\pi$.
So the integral that gives the volume is computed in spherical coordinates as follows. Note the "extra" factor of $\rho^2 * \text{Sin}[\phi]$ in the integrand.

$$\text{volume7} = \int_0^{2\pi} \int_{\pi/4}^{\pi/2} \int_0^1 \rho^2 * \text{Sin}[\phi] \, d\rho \, d\phi \, d\theta$$

$$\frac{\sqrt{2}\,\pi}{3}$$

```
N[volume7]
```

```
1.48096
```

Example 2.8. Find the volume of the region inside the spherical surface $\rho = 1 + \text{Sin}[5\,\theta]/5$, $0 \le \phi \le \pi$, $0 \le \theta \le 2\pi$.

Here is a picture of the solid.

```
starFruit = SphericalPlot3D[1 + Sin[5 θ] / 5, {ϕ, 0, π}, {θ, 0, 2 π},
          PlotStyle → Directive[Orange, Opacity[0.7], Specularity[White, 10]],
          Mesh → None, PlotPoints → 100,
          {Axes → True, Boxed → True}, ImageSize → 200]
```



The solid is easier to describe in spherical coordinates as : $0 \leq \rho \leq 1 + \text{Sin}[5\,\theta]/5,\ 0 \leq \phi \leq \pi,\ 0 \leq \ \leq 2\,\pi$.
The integral that gives the volume is computed in spherical coordinates as follows.

$$\text{volume8} = \int_0^{2\pi}\int_0^{\pi}\int_0^{1+\text{Sin}[5\,\theta]/5} \rho^2 * \text{Sin}[\phi]\, d\rho\, d\phi\, d\theta$$

$$\frac{106\,\pi}{75}$$

```
N[volume8]
```

```
4.44012
```

Solids of Revolution

Example 2.9. Find the volume of the the torus obtained by rotating about the *z* -axis the disk in the xz-plane with center (2, 0, 0) and the radius 1.

Here is a picture of the torus.

```
torus = ParametricPlot3D[
  {(2 + Cos[α]) Cos[θ], (2 + Cos[α]) Sin[θ], Sin[α]}, {α, 0, 2 π}, {θ, 0, 2 π},
      Mesh → None, PlotPoints → 50,
      {Axes → True, Boxed → True}, ImageSize → 200]
```



To find the volume, we view the torus as a solid of revolution about the *z*-axis. Each cross section is a unit disk.

Note: We use the Washer method: $V = \int_c^d \pi\left(R^2 - r^2\right) dz$.

```
volumeTorus = 2 ∫_0^1 π ((2 + √(1 - z^2))^2 - (2 - √(1 - z^2))^2) dz
```

$4\pi^2$

Example 2.10. Find the volume of the solid that remains when a hole of radius 2 is drilled through the center of a sphere of radius 4.

Here is a picture of the solid.

```
ParametricPlot3D[{{√(16 - z^2) * Cos[θ], √(16 - z^2) * Sin[θ], z},
  {2 * Cos[θ], 2 * Sin[θ], z}}, {θ, 0, 2 π}, {z, -√12, √12},
 Mesh → None,
 {Axes → True, Boxed → True},
 ImageSize → 200]
```



To find the volume, we view the solid as a solid of revolution about the z-axis and we use the Washer method: $V = \int_c^d \pi\left(R^2 - r^2\right) dz$.

```
volume10 = 2 ∫_0^√12 π (4^2 - (√(16 - z^2))^2) dz
```

$16\sqrt{3}\,\pi$

2.2. Exercises

This section provides you with ten exercise problems. For the first nine problems, you are asked to compute volumes or surface areas for some solids given in 3D space. For the last problem, you are asked to compute the volume(s) of your own 3D solid(s) created in Part I Problem 1.10.

Problem 2.1. Compute the volume of the solid that lies under the surface $z = x^2 y^2$ and over the region described by $0 \le x \le \text{Sin}[y]$ and $0 \le y \le \pi$. (This is the same solid as in Part I Problem 1.1.)

Problem 2.2. Compute the volume of the solid that occupies the region that lies under the plane $z = x - y/4$ and above the planar region between the graphs $y = \pm\text{Sin}[\pi x]$ for $0 \le x \le 1$. (This is the same solid as in Part I Problem 1.2.)

Problem 2.3. Compute the volume of the solid that lies under the surface $z = 1 - 4x^2 y^2$ and over the unit disk, i.e. the region inside the unit circle. (This is the same solid as in Part I Problem 1.3.)

Problem 2.4. Compute the area of the part of the surface $z = 1 - 4x^2 y^2$ that lies above the unit disk. (This is the top of the solid in the previous problem, i.e. Problem 2.3.)

Example 2.5. Find the area of the surface created when the cylinder $y^2 + z^2 = 1$ intersects the cylinders: $x^2 + z^2 = 1$. (This is the boundary of the solid in Example 2.3.)

Problem 2.6: Find the volume of the solid that lies under the surface $z = 5y$ and above the sector of the unit disk in which $0 \le \theta \le \pi$. (This is the same solid as in Part I Problem 1.4.)

Problem 2.7. A Bumpy Sphere: Find the volume of the solid occupies the region inside the spherical surface $\rho = 1 + (\text{Sin}[5\phi] * \text{Sin}[3\theta])/5$, $0 \le \phi \le \pi$, $0 \le \theta \le 2\pi$. (This is the same solid as in Part I Problem 1.5.)

Problem 2.8. Find the volume of the solid that lies in the intersection of the two spheres given by the equations: $x^2 + y^2 + z^2 = 1$ and $x^2 + y^2 + (z-1)^2 = 1$. (This is the same solid as in Part I Problem 1.6.)

Problem 2.9. Find the volume of the solid that lies below the sphere $x^2 + y^2 + z^2 = 1$ and above the cone $z = \sqrt{x^2 + y^2}$. (This is the same solid as in Part 1 Problem 1.7.)

Problem 2.10. Find the volume of the solid(s) you have created in Part I Problem 1.10.